# $Id: PKGBUILD 158970 2012-05-13 20:58:29Z jgc $
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=libcaca
pkgver=0.99.beta18
pkgrel=1
pkgdesc="Color AsCii Art library"
arch=('i686' 'x86_64')
license=('LGPL')
url="http://caca.zoy.org/wiki/libcaca"
depends=('imlib2' 'ncurses')
options=(!libtool !emptydirs)
source=(http://libcaca.zoy.org/files/libcaca/${pkgname}-${pkgver}.tar.gz)
md5sums=('93d35dbdb0527d4c94df3e9a02e865cc')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr --mandir=/usr/share/man \
    --enable-shared --disable-doc \
    --disable-cxx --disable-gl --disable-csharp
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
